PSpice Model 6CA7/EL34

INTRODUCTION

This document supplements the PSpice model 6CA7.INC, and provides some background operation to the operation of the model along with details of functionality modelled or not modelled as the case may be.

Whilst every care has been taken to duplicate the functionality of the modelled device as described here, it should be stressed that modelling is not a substitution for breadboarding or other prototyping methods.

MODELLED FUNCTIONS

Inter-electrode capacitance, screen current, grid current to a limited degree. The grid current is an approximation, and does not take into account grid current rise at low values of Va and Vs.

FUNCTIONS NOT MODELLED

A heater model is not implemented at this stage.

MODEL PERFORMANCE



Figure 1: Anode current

Figure 1 above shows anode current against swept anode voltage for a range of grid voltages between -40V and 0V in steps of 5V. Screen voltage in this instance is 350V.

The next diagram shows screen current overlaid on the anode current chart. Grid voltage and screen voltages are the same as the previous diagram.



Figure 2: Anode and screen current together

As can be clearly seen, the screen current rises sharply Va is reduced.

MODEL DESCRIPTION

The following describes the various components of the model and their interaction:

Eat — is the arctangent calculation which causes the fall off in emission at lower anode voltages.

Egs — is the emission contribution from the grid and screen, g_1 and g_2.

Egs2 — is Egs after raising to the power of 3/2 and factored by a constant so that it may be turned directly into a current value.

Ecath — is the cathode current value. This is the current between anode and cathode, although some of this may be diverted by the screen grid. Basically consists of Egs2 multiplied by Eat.

Ga — is the actual cathode current. Synonymous with Ecath.

Escrn — is the screen current value.

Gs is the actual screen current. Synonymous with Escrn.

Gg is the grid current value. This is an approximation at present, and will be improved as more information becomes available.

ALTERING THE MODEL FOR OTHER SIMULATORS

It may be necessary to use the model with other simulators, such as Berkeley SPICE 3f4, in which case some of the PSpice specific items will need to be altered.

The PSpice LIMIT{a,b,c} statement can, in instances where b is zero, be replaced by the SPICE 3f4 statement URAMP(a). Where LIMIT{a,b,c} is used, with b=0 and c=variable, the SPICE 3f4 statement U(a/c)*c can be used.

Duncan Munro
11th May 1997

Web page: http://www.duncanamps.com/
Forum: http://forum.duncanamps.com/
Email: postmaster@duncanamps.com

MODEL LISTING

```
*
* PSpice Subcircuit for 6CA7/EL34 output pentode
*
* Supported:   screen current and interelectrode capacitances, also grid
*              current to a limited degree.
*
* Unsupported: Heater model.
*
* Note that the grid current is guesswork on my part in the absence
* of any data...
*
* D.Munro - 02/05/97
*
* 02/05/97  Initial model.
*
*
* Pins   A  Anode
*        S  Screen
*        G  Grid
*        K  Cathode
*
.SUBCKT 6CA7 A S G K
*
* Calculate contribution to cathode current
*
Eat     at      0       VALUE={0.636*ATAN(V(A,K)/23)}
Egs     gs      0       VALUE={LIMIT{V(S,K)/9.3+V(G,K)*0.95,0,1E6}}
Egs2    gs2     0       VALUE={PWRS(V(gs),1.5)}
Ecath   cc      0       VALUE={V(gs2)*V(at)}
*
* Calculate anode current
*
Ga      A       K       VALUE={1.86E-3*V(cc)}
*
* Calculate screen current
*
Escrn   sc      0       VALUE={V(gs2)*(1.1-V(at))}
Gs      S       K       VALUE={1.518E-3*V(sc)}
*
* Grid current (approximation - does not model low va/vs)
*
Gg      G       K       VALUE={PWR(LIMIT{V(G,K)+1,0,1E6},1.5)*50E-6}
*
* Capacitances
*
Cg1     G       K       15.4p
Cak     A       K       8.4p
Cg1a    G       A       1.1p

.ENDS
```